
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.10 which is part of the ‘IMS_36523_IWD_12wk42’ test suite in ATS ‘iwd-EUTRA-B2012-03_D12wk42’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_10_IMS
Test Group:
IMS_34229\8\IMS_CC_RegistrationTestcases.ttcn

ATS Version:
iwd-EUTRA-B2012-03_D12wk42
System Simulator used:
R&S CMW500 
UE used:
Qualcomm MSM 8960
Verification Status:
PASS


4. Corrections required for IMS Registration test case 8.10
This section describes the TTCN changes required to make IMS registration test case 8.10 run correctly with a LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2012-03_D12wk42’ release.

Change 1 – Corrections to function ‘f_IMS_IpAddrMatch()’

	Function name
	f_IMS_IpAddrMatch()

	Reason for change
	Incorrect matching of IP address to received address in SIP header. 

	Summary of change
	Corrected index of Starting of brackets.

	TTCN module
	CommonIMS\IMS_CommonFunctions.ttcn

	MCC160 Comment
	


Before change

…
function f_IMS_IpAddrMatch(IP_AddrInfo_Type p_AddrInfo,

                             charstring p_ReceivedAddrString) return boolean

  {

    var charstring v_AddrString;

    var integer v_Length;

    var integer i;

    var integer n;

    if (ischosen(p_AddrInfo.V4)) {

      return match(p_ReceivedAddrString, p_AddrInfo.V4.Addr);

    }

    if (ischosen(p_AddrInfo.V6)) {

      v_AddrString := p_ReceivedAddrString;

      v_Length := lengthof(v_AddrString);

      for (i := 0; i < v_Length; i := i + 1) {                                  // get rid of any "[" etc. at the beginning of the string

        if (match(v_AddrString[i], pattern "[0-9:]")) { break; }

      }

      for (n := v_Length; n > 0; n := n - 1) {                                  // get rid of any "]" etc. at the end of the string

        if (match(v_AddrString[n-1], pattern "[0-9:]")) { break; }

      }

      v_AddrString := substr(v_AddrString, i, n - i);

      return (f_Convert_IPv6Addr2OctString(v_AddrString) == f_Convert_IPv6Addr2OctString(p_AddrInfo.V6.Addr));

    }

    return false;

}
…

After change

…

  function f_IMS_IpAddrMatch(IP_AddrInfo_Type p_AddrInfo,

                             charstring p_ReceivedAddrString) return boolean

  {

    var charstring v_AddrString;

    var charstring v_AddrStringRouted;

    var integer v_Length;

    var integer v_LengthRouted;

    var integer i;

    var integer n;

    if (ischosen(p_AddrInfo.V4)) {

      return match(p_ReceivedAddrString, p_AddrInfo.V4.Addr);

    }

    if (ischosen(p_AddrInfo.V6)) {

      v_AddrString := p_ReceivedAddrString;

       v_Length := lengthof(v_AddrString);

      for (i := 0; i < v_Length; i := i + 1) {                                  // get rid of any "[" etc. at the beginning of the string

        if (match(v_AddrString[i], pattern "[a-z]|[0-9:]")) { break; }

      }

      for (n := v_Length; n > 0; n := n - 1) {                                  // get rid of any "]" etc. at the end of the string

        if (match(v_AddrString[n-1], pattern "[0-9:]")) { break; }

      }

      v_AddrString := substr(v_AddrString, i, n - i);

      return (f_Convert_IPv6Addr2OctString(v_AddrString) == f_Convert_IPv6Addr2OctString(p_AddrInfo.V6.Addr));

    }

…

Change 2 – Corrections to template ‘cr_ViaInitialReg’

	Template name
	cr_ViaInitialReg

	Reason for change
	For initial GIBA registration, the via header is not required to contain port number (see 34.229-1 A.1.1 condition for GIBA registration).

	Summary of change
	Port number in via header not checked as mandatory present.

	TTCN module
	CommonIMS\IMS_SIP_Templates.ttcn

	MCC160 Comment
	


Before change

…

  template Via cr_ViaInitialReg(template (present) charstring p_Transport := ?,

                         template (present) HostPort p_HostPort := cr_HostPort(?, *)) :=

  {

    fieldName := VIA_E,

    viaBody := {

      cr_ViaBodyInitialReg(p_Transport, p_HostPort)

    }

  };
…

After change

…

  template Via cr_ViaInitialReg(template (present) charstring p_Transport := ?,

                         template (present) HostPort p_HostPort := cr_HostPort(?, *)) :=

  {

    fieldName := VIA_E,

    viaBody := {

      cr_ViaBody(p_Transport, p_HostPort)

    }

  };

…  

Change 3 – Corrections to local function ‘fl_IMS_ContactAddress_CheckParams()’

	Function name
	fl_IMS_ContactAddress_CheckParams()

	Reason for change
	For GIBA registration, see Annex A.1.1. condition A3, the feature parameters in the Contact header do not have to be checked as these parameters are optional.

	Summary of change
	Check commented out.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

function fl_IMS_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                             boolean p_CondA7 := false,

                                             boolean p_IMS_MTSI := tsc_IMS_MTSI,

                                             boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                             boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS) runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      if (not match(v_ContactParams, cr_ReqTypeParamInContactParams("sos"))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_MTSI) {                                                             // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_ContactFeatureParam))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }
    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

}
…

After change

…

function fl_IMS_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                             boolean p_CondA7 := false,

                                             boolean p_IMS_MTSI := tsc_IMS_MTSI,

                                             boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                             boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS) runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      if (not match(v_ContactParams, cr_ReqTypeParamInContactParams("sos"))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    // Rel-7 specific, @sic R5-073440 sic


//CR needed: This value is not mandatory should be a pixit
/* NOT needed for GIBA registration

    if (p_IMS_MTSI) {                                                             // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_ContactFeatureParam))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

*/
    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

  }   

…  

Change 4 – Corrections to function ‘f_IMS_RegisterRequest_MessageHeaderRX_Initial()’

	Function name
	f_IMS_RegisterRequest_MessageHeaderRX_Initial()

	Reason for change
	For GIBA registration, see Annex A.1.1. condition A3, User info was expected as omit in the Contact header. User Information is not mandatory and can be omítted.

	Summary of change
	Added and used new template ‘cr_SipUrl_HostPortUserInfo’.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

  function f_IMS_RegisterRequest_MessageHeaderRX_Initial(boolean p_CondA3 := false) runs on IMS_PTC return template (present) MessageHeader

  { /* initial REGISTER (condition A1 or A3) acc. to 34.229-1 Annex A.1.1

       p_CondA3  .. condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA)

       NOTE: several checks are done explicitly in the respective check functions applied on the receive messages (the same for sigcomp) */

    var template (present) SipUrl v_SipUrlPublicUserId := ?;   /* Acc. to 24.229 cl. 5.1.1.2.1 for initial registration the UE shall populate

                                                                  'To' and 'From' with the public user identity to be registered;

                                                                  Normally this is one of the IMPUs stored on the ISIM;

                                                                  The public user identity is explicitly checked after matching the REGISTER message */

    var template (present) integer v_CseqValue := ?;

    var template (present) MessageHeader v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(p_CondA3, v_CseqValue);

    v_MessageHeader_Register.via           := cr_ViaInitialReg;

    v_MessageHeader_Register.route       := cr_RouteRegDef(px_IMS_Pcscf, *) ifpresent;

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPort(?, *));  // checked in fl_IMS_ContactAddress_CheckParams, f_IMS_A_1_1_MessageHeader_CommonChecks

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *) ifpresent;    // optional for A1, A3; cellId is not checked

    if (not p_CondA3) {  // => 34.229-1 A.1.1 condition A1

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := cr_AuthorizationInitial(px_IMS_Private_UserId, px_IMS_HomeDomainName);

    }

    return v_MessageHeader_Register;

  }      

…

After change

…

  function f_IMS_RegisterRequest_MessageHeaderRX_Initial(boolean p_CondA3 := false) runs on IMS_PTC return template (present) MessageHeader

  { /* initial REGISTER (condition A1 or A3) acc. to 34.229-1 Annex A.1.1

       p_CondA3  .. condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA)

       NOTE: several checks are done explicitly in the respective check functions applied on the receive messages (the same for sigcomp) */

    var template (present) SipUrl v_SipUrlPublicUserId := ?;   /* Acc. to 24.229 cl. 5.1.1.2.1 for initial registration the UE shall populate

                                                                  'To' and 'From' with the public user identity to be registered;

                                                                  Normally this is one of the IMPUs stored on the ISIM;

                                                                  The public user identity is explicitly checked after matching the REGISTER message */

    var template (present) integer v_CseqValue := ?;

    var template (present) MessageHeader v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(p_CondA3, v_CseqValue);

    v_MessageHeader_Register.via           := cr_ViaInitialReg;

    v_MessageHeader_Register.route       := cr_RouteRegDef(px_IMS_Pcscf, *) ifpresent;

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPortUserInfo(?, *));  // checked in fl_IMS_ContactAddress_CheckParams, f_IMS_A_1_1_MessageHeader_CommonChecks

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *) ifpresent;    // optional for A1, A3; cellId is not checked

    if (not p_CondA3) {  // => 34.229-1 A.1.1 condition A1

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := cr_AuthorizationInitial(px_IMS_Private_UserId, px_IMS_HomeDomainName);

    }

    return v_MessageHeader_Register;

  }
……………………………………….


template SipUrl cr_SipUrl_HostPortUserInfo(template charstring p_Host,

                                     template integer p_Port := omit) :=

    cr_SipUrl(-, *, p_Host, p_Port);
…  

Change 5 – Corrections to function ‘f_IMS_SubscribeRequest_MessageHeaderRX()’

	Function name
	f_IMS_SubscribeRequest_MessageHeaderRX

	Reason for change
	For GIBA registration, see Annex A.1.1. condition A3, User info was expected as omit in the Contact header. User Information is not mandatory and can be omítted.

	Summary of change
	Used new template introduced in change 4 above..

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

function f_IMS_SubscribeRequest_MessageHeaderRX(boolean p_CondA2 := false,

                                                  EventType p_EventType := "reg") runs on IMS_PTC return template (present) MessageHeader

  { /* SUBSCRIBE acc. to 34.229-1 Annex A.1.4 */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) SipUrl v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetPublicUserId_SipUrl_RX();

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Subscribe.expires     := cr_Expires("600000");

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (p_CondA2) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, tsc_IMS_PortNumber_5060));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_Require;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef ("application/reginfo+xml") ifpresent;

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

  }
…

After change

…

function f_IMS_SubscribeRequest_MessageHeaderRX(boolean p_CondA2 := false,

                                                  EventType p_EventType := "reg") runs on IMS_PTC return template (present) MessageHeader

  { /* SUBSCRIBE acc. to 34.229-1 Annex A.1.4 */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) SipUrl v_SipUrlPublicUserId := f_IMS_PTC_ImsInfo_GetPublicUserId_SipUrl_RX();

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlPublicUserId);

    v_MessageHeader_Subscribe.expires     := cr_Expires("600000");

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    //CR needed:UserInfo not omitted

    if (p_CondA2) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPortUserInfo(?, tsc_IMS_PortNumber_5060));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPortUserInfo(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_Require;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef ("application/reginfo+xml") ifpresent;

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

}
…  

Change 6 – Corrections to function ‘f_IMS_Register_Subscribe()’

	Function name
	f_IMS_Register_Subscribe ()

	Reason for change
	NW address and UE address were missing when SS sent SIP Notify message.

	Summary of change
	Return value modified.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

function f_IMS_Register_Subscribe(boolean p_CondA2 := false) runs on IMS_PTC return SUBSCRIBE_Request
  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    v_IMS_DATA_REQ := f_IMS_SUBSCRIBE_Request(p_CondA2);

    f_IMS_SUBSCRIBE_Response(v_IMS_DATA_REQ, p_CondA2);

    return v_IMS_DATA_REQ.Request.Subscribe;

}
…

After change

…

function f_IMS_Register_Subscribe(boolean p_CondA2 := false) runs on IMS_PTC return IMS_DATA_REQ
  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    v_IMS_DATA_REQ := f_IMS_SUBSCRIBE_Request(p_CondA2);

    f_IMS_SUBSCRIBE_Response(v_IMS_DATA_REQ, p_CondA2);

    return v_IMS_DATA_REQ;
  }

…  

Change 7 – Corrections to function ‘f_IMS_Register_Notify()’

	Function name
	f_IMS_Register_Notify ()

	Reason for change
	NW address and UE address were missing when SS sent SIP Notify message.

	Summary of change
	New input parameter added.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

function f_IMS_Register_Notify(SipUrl p_ContactUrl,

                                 SUBSCRIBE_Request p_SubscribeRequest,

                                 template (value) Reginfo p_XmlMessage,

                                 boolean p_CondA2 := false,

                                 integer p_CseqValue := 1,

                                 template (value) SubscriptionState p_SubscriptionState := cs_SubscriptionState_Active) runs on IMS_PTC

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var IMS_ProtectedUnprotected_Type v_ProtectedUnprotected;

    var template (value) MessageHeader v_MessageHeader_Notify;

    var template (present) MessageHeader v_MessageHeader_Response;

    if (not p_CondA2) { v_ProtectedUnprotected := protected; }

    else {              v_ProtectedUnprotected := unprotected; }

    v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A16(p_SubscribeRequest, p_CondA2, p_CseqValue, p_SubscriptionState);

    IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(v_ProtectedUnprotected, f_IMS_PTC_ImsInfo_GetTransportProtocol()),

                                           cs_NOTIFY_Request(p_ContactUrl, v_MessageHeader_Notify, cs_MessageBody_RegInfo(p_XmlMessage))));

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtectedUnprotected, f_IMS_PTC_ImsInfo_GetTransportProtocol()),

                                        cr_Response(c_statusLine200, v_MessageHeader_Response)));

}
…

After change

…

function f_IMS_Register_Notify(SipUrl p_ContactUrl,

                                 SUBSCRIBE_Request p_SubscribeRequest,

                                 template (value) Reginfo p_XmlMessage,

                                 boolean p_CondA2 := false,

                                 integer p_CseqValue := 1,

                                 template (value) SubscriptionState p_SubscriptionState := cs_SubscriptionState_Active,






   template (value) IMS_RoutingInfo_Type p_RoutingInfo:= cs_IMS_RoutingInfo(unprotected, f_IMS_PTC_ImsInfo_GetTransportProtocol())) runs on IMS_PTC

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var IMS_ProtectedUnprotected_Type v_ProtectedUnprotected;

    var template (value) MessageHeader v_MessageHeader_Notify;

    var template (present) MessageHeader v_MessageHeader_Response;

    if (not p_CondA2) { v_ProtectedUnprotected := protected; }

    else {              v_ProtectedUnprotected := unprotected; }

    v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A16(p_SubscribeRequest, p_CondA2, p_CseqValue, p_SubscriptionState);

    IMS_Client.send(cas_IMS_Notify_Request(p_RoutingInfo,

                                           cs_NOTIFY_Request(p_ContactUrl, v_MessageHeader_Notify, cs_MessageBody_RegInfo(p_XmlMessage))));

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);

    IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(v_ProtectedUnprotected, f_IMS_PTC_ImsInfo_GetTransportProtocol()),

                                        cr_Response(c_statusLine200, v_MessageHeader_Response)));

}
…  

Change 8 – Corrections to function ‘f_IMS_Register_SubscribeNotify()’

	Function name
	f_IMS_Register_SubscribeNotify()

	Reason for change
	NW address and UE address were missing when SS sent SIP Notify message.

	Summary of change
	Correct Routing Info parameter added..

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

function f_IMS_Register_SubscribeNotify(REGISTER_Request p_RegisterRequest,

                                          boolean p_IsGIBA := false) runs on IMS_PTC return SUBSCRIBE_Request

  {

    var SUBSCRIBE_Request v_SubscribeReq;

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterRequest.msgHeader);

    // Steps 8 and 9

    v_SubscribeReq := f_IMS_Register_Subscribe(p_IsGIBA);

    // Steps 10  and 11

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, f_IMS_XmlMessageForNotify_Def(p_RegisterRequest), p_IsGIBA);
    return v_SubscribeReq;

}
…

After change

…

function f_IMS_Register_SubscribeNotify(REGISTER_Request p_RegisterRequest,

                                          boolean p_IsGIBA := false) runs on IMS_PTC return SUBSCRIBE_Request

  {

    var SUBSCRIBE_Request v_SubscribeReq;


var IMS_DATA_REQ v_DataReq;

    var SipUrl v_ContactUrl := f_MessageHeader_GetContactSipUrl(p_RegisterRequest.msgHeader);

    // Steps 8 and 9

    v_DataReq := f_IMS_Register_Subscribe(p_IsGIBA);
    v_SubscribeReq := v_DataReq.Request.Subscribe;

    // Steps 10  and 11

    f_IMS_Register_Notify(v_ContactUrl, v_SubscribeReq, f_IMS_XmlMessageForNotify_Def(p_RegisterRequest), p_IsGIBA, -,-,cs_IMS_RoutingInfo(unprotected, f_IMS_PTC_ImsInfo_GetTransportProtocol(),v_DataReq.RoutingInfo.UE_Address,v_DataReq.RoutingInfo.NW_Address));

    return v_SubscribeReq;

}
…  

Change 9 – Corrections to step ‘a_IP_IMS_UL_ReceiveData’

	Function name
	a_IP_IMS_UL_ReceiveData

	Reason for change
	UL Data Response was incorrectly handled by the IP PTC in unprotected case.

	Summary of change
	Routing info handled dynamically for UL Data Requests and Responses.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

…

  altstep a_IP_IMS_UL_ReceiveData(integer p_PdnIndex,

                                  template (present) IP_AddrInfo_Type p_IpAddr,

                                  template (present) PortNumber_Type  p_PortNumber,

                                  IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive UDP datagram from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive unprotected messages

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;

        select (p_ProtectedUnprotected) {

          case (unprotected) {

            /* initial access: unprotected connection; => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

            v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    // may be IPv4 or IPv6

            v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   // may be IPv4 or IPv6

            v_ClientServer := server;
          }

          case (protected) {

            v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                         * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                         * cannot be distinguished by the port on which they have been received */

          }

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(p_ProtectedUnprotected, udp, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

}
…

After change

…

altstep a_IP_IMS_UL_ReceiveData(integer p_PdnIndex,

                                  template (present) IP_AddrInfo_Type p_IpAddr,

                                  template (present) PortNumber_Type  p_PortNumber,

                                  IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive UDP datagram from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive unprotected messages

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;

        select (p_ProtectedUnprotected) {

          case (unprotected) {

            /* initial access: unprotected connection; => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

            v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    // may be IPv4 or IPv6

            v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   // may be IPv4 or IPv6

            v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);
          }

          case (protected) {

            v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                         * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                         * cannot be distinguished by the port on which they have been received */

          }

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(p_ProtectedUnprotected, udp, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

}
…  

Change 10 – Corrections to function ‘f_TC_8_10_IMS()’

	Function name
	f_TC_8_10_IMS()

	Reason for change
	Pass verdict was not assigned at the end of the TestBody part.

	Summary of change
	Verdict assignment added.

	TTCN module
	8\ IMS_CC_RegistrationTestcases.ttcn

	MCC160 Comment
	


Before change

…

  // Test case 8.10  -  See 34.229-1 clause 8.10

  function f_TC_8_10_IMS() runs on IMS_PTC

  {

    var boolean v_IsGIBA := true;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Steps 1, 2 and 3 to 6

    f_IMS_Registration(v_IsGIBA);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

}
…

After change

…

  // Test case 8.10  -  See 34.229-1 clause 8.10

  function f_TC_8_10_IMS() runs on IMS_PTC

  {

    var boolean v_IsGIBA := true;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Steps 1, 2 and 3 to 6

    f_IMS_Registration(v_IsGIBA);


    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step6");
    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }
…  

5. Execution Log Files

5.1 Qualcomm MSM 8960 UE 

The Qualcomm MSM 8960 UE passed this IMS Registration test case over LTE on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 11. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_8_10_IMS_LTE_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s120859:    Supporting information for agreement of IMS Registration TC 8.10 over LTE. 
This archive comprises:

                        - text format execution log files

